
Dans la lignée du journal EDSL et F-algèbres, ce journal présente une méthode pour plonger un langage dans un autre (ici OCaml) qui généralise la précédente et centrée autour de la notion d'interprétation. Contrairement aux méthodes plus courantes pour résoudre cette question, la méthode tagless-final permet également de résoudre le problème de l'extensibilité : étendre un type de donnés, ajouter des opérations dessus, sans avoir à réécrire du code déjà compilé et cela avec la sécurité du typage statique.


J'ai essayé d'écrire le journal de telle façon qu'il ne soit pas nécessaire de connaître les fondamentaux du paradigme fonctionnel en programmation, même si une familiarité avec d'autres langages que OCaml sont indispensables. Si certains points vous semblent trop obscurs ou mal expliqués, les commentaires sont là pour vos questions.


Avant de présenter l'approche tagless-final et son originalité, je commencerai par présenter la façon plus « standard » de procéder. Celle-ci consistant à manipuler des structures de données, elle est moins spécifique au langage fonctionnel. Elle sera sans doute plus facile à appréhender dans un premier temps, et permettra de servir de point de comparaison.

Approche par AST comme structure de données ou méthode initiale.


Dans cette première partie, nous étudierons la manière usuelle de représenter les termes d'un langage via une structure de données correspondant à son AST (arbre de syntaxe abstrait). Nous commencerons en présentant cette méthode sur un langage algébrique simple d'expressions sur des entiers.

Langage additif et multiplicatif sur les entiers.


Dans un premier temps, nous allons voir comment représenter en OCaml un langage de calcul sur les entiers avec des expressions comme 1 + 2 + 3, ou 1 + 2 * 3, ou encore (1 + 2) * (3 + 4). La technique courante pour traiter un tel langage en OCaml est de définir un type de données représentant les termes du langage :


type expr = 
  | Lit : int -> expr (* un littéral représentant un entier *)
  | Add : expr * expr -> expr (* une expression représentant une addition *)
  | Mul : expr * expr -> expr (* une expression représentant une multiplication *)


Ce type de données est ce que l'on appelle un type somme ou variant, et ressemble un peu au union du C (en). Chaque ligne définit un constructeur pour ce type de données (par exemple le constructeur Add prend un couple de expr et renvoie un expr) et chaque valeur de ce type correspond à un et un seul de ces trois cas. Afin d'obtenir une expression, on utilise nos constructeurs comme dans ces exemples :


(* 1 + 2 *)
let ex1 = Add(Lit 1, Lit 2);;
val ex1 : expr = Add (Lit 1, Lit 2)

(* (1 + 2) * (3 + 4) *)
let ex2 = Mul(Add(Lit 1, Lit 2), Add(Lit 3, Lit 4));;
val ex2 : expr = Mul (Add (Lit 1, Lit 2), Add (Lit 3, Lit 4))


Une façon de se représenter visuellement une telle structure est sous la forme d'un arbre dont les nœuds sont étiquetés par les constructeurs Add ou Mul et où les feuilles sont des littéraux.


           Mul
         /     \
        /       \
       /         \
     Add         Add
    /   \       /   \
   /     \     /     \
Lit 1  Lit 2  Lit 3  Lit 4



Cette représentation est ce que l'on appelle l'arbre de syntaxe abstrait de l'expression qui, dans notre cas, est le produit de deux additions. Maintenant, afin d'évaluer la valeur entière de celle-ci, on va définir une fonction eval qui parcourt l'arbre récursivement pour le convertir en un entier.


let rec eval : expr -> int = function
  | Lit n -> n
  | Add (e1,e2) -> (eval e1) + (eval e2)
  | Mul (e1,e2) -> (eval e1) * (eval e2)


La définition de la fonction est simple et va de soi : on fait une étude de cas sur la forme que peut avoir notre expression, et on effectue le traitement adapté. Par exemple, dans le cas du constructeur Add on commence par évaluer récursivement chacune des sous-expressions e1 et e2 puis on en calcule la somme. Le compilateur se charge de vérifier que tous les cas possibles sont bien gérés et que l'analyse est exhaustive, sinon il émet un avertissement :


let rec eval : expr -> int = function
  | Lit n -> n
  | Add (e1,e2) -> (eval e1) + (eval e2);;
Characters 29-93:
Warning 8: this pattern-matching is not exhaustive.
Here is an example of a case that is not matched:
Mul (_,_)
val eval : expr -> int = <fun>


Ici on a oublié de traiter le cas de la multiplication. On peut maintenant tester notre fonction d'évaluation sur nos exemples :


(* 1 + 2 *)
eval ex1;;
- : int = 3

(* (1 + 2) * (3 + 4) *)
eval ex2;;
- : int = 21


On pourrait aussi, au lieu d'évaluer la valeur du terme, afficher l'arbre sous forme de chaîne de caractères exprimant le calcul à effectuer.


(* une fonction qui prend deux chaînes puis rajoute un + entre les deux *)
let show_add s s' = s ^ " + " ^ s'
(* la même mais pour la multiplication *)
let show_mul s s' = s ^ " * " ^ s'
(* une fonction qui met une chaîne entre paranthèses *)
let paren s = "(" ^ s ^ ")"

(* notre fonction d'affichage *)
let rec view : expr -> string = function
  | Lit n -> string_of_int n
  | Add (e1,e2) -> paren (show_add (view e1) (view e2))
  | Mul (e1,e2) -> paren (show_mul (view e1) (view e2))

let _ = view ex1;;
- : string = "(1 + 2)"

let _ = view ex2;;
- : string = "((1 + 2) * (3 + 4))"


Si l'on regarde le code des deux fonctions eval et view ci-dessus, l'on constate un motif commun :



	on décompose l'arbre en ses constituants et on applique une fonction aux éléments dans chaque cas ;

	la valeur d'un nœud ne dépend que de la valeur des ses sous-arbres.


On peut alors coder une fonction générique capturant ce motif qui porte traditionnellement le nom de fold (en) :


let rec fold lit add mul = fun e ->
  let eval_or_view = fold lit add mul in
  match e with
  | Lit i -> lit i
  | Add (e1,e2) -> add (eval_or_view e1) (eval_or_view e2)
  | Mul (e1,e2) -> mul (eval_or_view e1) (eval_or_view e2)

let eval = fold (fun i -> i) (+) ( * )
let view =
  fold string_of_int
       (fun s s' -> paren (show_add s s'))
       (fun s s' -> paren (show_mul s s'))


Dans la définition de fold, j'ai volontairement définie une fonction locale eval_or_view pour souligner la correspondance avec les codes précédents. Regardons un instant le type inféré pour ces trois fonctions :


val fold : (int -> 'a) -> ('a -> 'a -> 'a) -> ('a -> 'a -> 'a) -> expr -> 'a
val eval : expr -> int
val view : expr -> string


La fonction fold prend pour argument une fonction lit : int -> 'a, une fonction add : 'a -> 'a -> 'a et une fonction mul : 'a -> 'a -> 'a qui constituent les interprétations des littéraux et des opérateurs du langage. Le type 'a constitue le domaine dans lequel on interprète nos termes, à savoir int dans le cas de eval et string dans le cas de view. Le principe général de la fonction fold consiste alors à parcourir l'arbre et à remplacer récursivement chaque constructeur par l'application d'une fonction correspondant à l'interprétation de celui-ci.


Nous avons vu là les principes de base pour définir un langage en OCaml :



	on représente les termes du langage via un variant ;

	on code une fonction fold générique sur celui-ci ;

	on se sert de cette fonction pour faire varier les interprétations.


Nous allons voir, à présent, comment faire pour coder des interprétations dites contextuelles à travers l'exemple du pretty printing.

Interprétation contextualisée : le cas du pretty printing.


Dans cette section, nous allons traiter la notion d'interprétation dite contextualisée : l'interprétation d'un terme ne dépend plus alors seulement de ses constituants ou sous-termes, mais également du contexte dans lequel il est utilisé. Un exemple classique de telles interprétations est celui du pretty printing lorsque l'on cherche à afficher élégamment la structure de l'arbre syntaxique.


L'objectif ici sera d'afficher l'arbre en supprimant certaines parenthèses superflues. Si l'on reprend les deux exemples de la section précédente, nous avons pour l'instant :


view ex1;;
- : string = "(1 + 2)"

view ex2;;
- : string = "((1 + 2) * (3 + 4))"


Et l'on voudrait arriver à un affichage minimisant les parenthèses nécessaires à la représentation textuelle de la structure de l'arbre, les opérations étant vues comme associatives à gauche (ce qui est le cas en OCaml) :


show ex1;;
- : string = "1 + 2"

show ex2;;
- : string = "(1 + 2) * (3 + 4)"


Ici l'interprétation d'un terme sous forme de string dépend du contexte dans lequel il est employé : dans le cas de ex2 le terme 1 + 2 qui apparaît comme premier opérande de la multiplication est mis entre parenthèses, alors qu'il ne l'est pas dans le cas de ex1. Une solution consiste à utiliser un contexte p de type int représentant le niveau de précédence dans lequel se trouve le terme, puis de mettre des parenthèses en fonction de la valeur de celui-ci.


(* une fonction de parenthésage sous condition *)
let cparen b = if b then paren else (fun s -> s)

let rec cshow p e = match p,e with
  | _ , Lit n -> string_of_int n
  | p , Add (e1,e2) -> cparen (p > 3) @@ show_add (cshow 3 e1) (cshow 4 e2)
  | p , Mul (e1,e2) -> cparen (p > 4) @@ show_mul (cshow 4 e1) (cshow 5 e2)

cshow 0 ex1;;
- : string = "1 + 2"

cshow 0 ex2;;
- : string = "(1 + 2) * (3 + 4)"

(* avec un nouvel exemple : 1 + (2 + 3 * 4) *)
let ex3 = Add(Lit 1, Add(Lit 2, Mul(Lit 3, Lit 4)));;
val ex3 : expr = Add (Lit 1, Add (Lit 2, Mul (Lit 3, Lit 4)))

cshow 0 ex3;;
- : string = "1 + (2 + 3 * 4)"


Dans le code ci-dessus, on vient de rencontrer une nouvelle notation pour l'application de fonction avec l'opérateur infixe @@. Il est défini en OCaml par l'expression let (@@) f x = f x, ce qui signifie que les expressions f @@ x et f x sont équivalentes. Il permet d'éviter d'avoir recours à de trop nombreuses parenthèses — contrairement au LISP par exemple — autour des arguments d'une fonction, et de faciliter ainsi la lecture et la compréhension du code.


Le motif de la fonction cshow : int -> expr -> string ne suit malheureusement pas celui générique du fold : l'interprétation d'un terme ne dépend pas seulement de la valeur de ses sous-termes. Ici le paramètre p : int sert d'accumulateur quand on descend dans la structure et porte avec lui l'information pour savoir s'il faut ou non parenthèser l'expression — information non prise en compte pour les littéraux.


Que se passe-t-il si l'on définit la même fonction mais en inversant l'ordre des arguments ?


let rec cshow e p = match e,p with
  | Lit n , _ -> string_of_int n
  | Add (e1,e2) , p -> cparen (p > 3) @@ show_add (cshow e1 3) (cshow e2 4)
  | Mul (e1,e2) , p -> cparen (p > 4) @@ show_mul (cshow e1 4) (cshow e2 5)
val cshow : expr -> int -> string = <fun>


On commence à se rapprocher du motif de la fonction fold et la fonction prend bien une valeur de type expr pour renvoyer une fonction de type int -> string. Et si l'on pouvait se ramener à une interprétation via fold mais où l'on interpréterait notre arbre comme une fonction de type int -> string ?


Pour y arriver, faisons un petit détour par la curryfication de fonctions. Le principe est bien connu chez les adeptes de la programmation fonctionnelle, mais peut être moins chez les développeurs non habitués à ce paradigme. C'est surtout pour eux que le présent passage est écrit, pour les autres il suffit de le survoler.


Une fonction est caractérisée par un ensemble de définition (son domaine), un ensemble d'arrivée (son codomaine) et une relation univoque entre un élément du domaine vers un élément du codomaine. Lorsque l'on traite des fonctions de plusieurs variables, le domaine est constitué de n-uplets ou tuples. Le principe de la curryfication consiste à se ramener, dans ce cas, à des fonctions d'une seule variable. Illustrons la chose en python :


def plus (x, y):
  return x + y

>>> plus (1, 2)
3


Ici le domaine est constitué des couples d'entiers et le codomaine des entiers. La curryfication de cette fonction va consister à la transformer en une fonction dont le domaine est constitué des entiers et le codomaine des fonctions des entiers vers les entiers :


plus_curry = lambda x : lambda y : x + y
>>> plus_curry (1)
<function <lambda>.<locals>.<lambda> at 0x7f64b9d46f28>
>>> plus_curry (1) (2)
3


En OCaml, ces exemples donneraient :


let plus = fun (x, y) -> x + y
let plus_curry = fun x -> fun y -> x + y

plus (1, 2);;
- : int = 3

plus_curry 1;;
- : int -> int = <fun>

plus_curry 1 2;;
- : int = 3


Après cette légère digression, revenons à notre fonction de pretty printing :


let rec cshow e p = match e,p with
  | Lit n , _ -> string_of_int n
  | Add (e1,e2) , p -> cparen (p > 3) @@ show_add (cshow e1 3) (cshow e2 4)
  | Mul (e1,e2) , p -> cparen (p > 4) @@ show_mul (cshow e1 4) (cshow e2 5)
val cshow : expr -> int -> string = <fun>


On décompose le couple (e, p) et selon les cas l'on renvoie une chaîne de caractères particulière, il suffit donc de curryfier chaque branche de l'alternative pour obtenir :


let rec cshow e = match e with
  | Lit n -> fun p -> string_of_int n
  | Add (e1,e2) -> fun p ->
      cparen (p > 3) @@ show_add (cshow e1 3) (cshow e2 4)
  | Mul (e1,e2) -> fun p ->
      cparen (p > 4) @@ show_mul (cshow e1 4) (cshow e2 5)
val cshow : expr -> int -> string = <fun>


Ce qui, au final, donne la version suivante de show à base de fold :


let show e =
  let lit i = fun  p -> string_of_int i in
  let add x y = fun p -> cparen (p > 3) @@ show_add (x 3) (y 4) in
  let mul x y = fun p -> cparen (p > 4) @@ show_mul (x 4) (y 5) in
  fold lit add mul e 0

show ex2;;
- : string = "(1 + 2) * (3 + 4)"

show ex3;;
- : string = "1 + (2 + 3 * 4)"


Nous avons vu que même dans le cas où l'interprétation semblait devoir dépendre d'un contexte, l'on pouvait se ramener au cas d'une interprétation sans contexte en changeant le domaine d'interprétation via la curryfication. Essayons à présent d'enrichir un peu notre langage en lui rajoutant une structure conditionnelle du genre if-then-else.

Enrichissons le langage avec des booléens et conditions.


La première idée qui vient à l'esprit est de faire comme précédemment en définissant un variant pour représenter notre nouveau langage. Pour la structure if-then-else, on placera chaque branche dans un thunk (en) car le langage OCaml fait de l'appel par valeur, ainsi l'expression ne sera évaluée que si l'on en a réellement besoin.


type expr =
  | Lit : int -> expr
  | Add : expr * expr -> expr
  | Mul : expr * expr -> expr
  | Bool : bool -> expr
  | If : expr * (unit -> expr) * (unit -> expr) -> expr


Voilà qui est bien mais cela pose deux problèmes :



	ce nouveau type n'est pas compatible avec le premier et il faut réécrire toutes nos fonctions ;

	il permet de construire des expressions mal typées, comme ajouter deux booléens.


Le premier point peut être géré, entre autre, via ce que l'on appelle les variants polymorphes mais cette solution, que je ne traiterai pas, ne permet pas de résoudre le second point. Je vais néanmoins montrer comment résoudre le second grâce aux GADT ou types algébriques généralisés. En revanche le premier problème persistera : il faudra réécrire nos fonctions.


Un GADT consiste à utiliser un type paramétré par un autre type et à s'en servir pour typer tant les paramètres que la sortie de chacun des constructeurs du variant.


type _ expr =
  | Lit : int -> int expr
  | Add : int expr * int expr -> int expr
  | Mul : int expr * int expr -> int expr
  | Bool : bool -> bool expr
  | If : bool expr * (unit -> 'a expr) * (unit -> 'a expr) -> 'a expr


En fixant les bonnes valeurs du paramètre de type dans chacune des branches, on s'assure ainsi de ne jamais pouvoir ajouter deux booléens sans déclencher d'erreurs de typage à la compilation :


Add (Bool true, Bool false);;
     ^^^^^^^^^
Error: This expression has type bool expr
but an expression was expected of type int expr
Type bool is not compatible with type int


Outre la garantie de typage qu'ils offrent pour notre langage, les GADT sont un outil puissant afin de garantir de nombreux invariants sur des structures de données. Leurs possibilités dépassent de loin le cadre de ce journal, mais on pourra trouver un exposé détaillé dans cette leçon en anglais accompagné de son TD qui permet d'utiliser OCaml dans son navigateur.


Le typage de fonctions impliquant des GADT n'est en général pas décidable et il n'est pas possible de définir une fonction aussi générique que le fold comme pour les variants classiques. Je me contenterai simplement de montrer comment écrire la fonction eval pour ce nouveau langage.


let rec eval : type a. a expr -> a = function
  | Lit n -> n
  | Add (x,y) -> (eval x) + (eval y)
  | Mul (x,y) -> (eval x) * (eval y)
  | Bool b -> b
  | If (b,th,el) -> if (eval b) then eval (th ()) else eval (el ())


Il faut ajouter une annotation de typage avec un type local abstrait (type a) pour préciser au compilateur que le type de retour est le même que le paramètre de type de l'expression. En dehors de cette annotation, le corps de la fonction est équivalente à sa version avec variant classique. Son retour sur un exemple :


eval (If (Bool true, (fun () -> Lit 1), (fun () -> Lit 2)));;
- : int = 1

Approche par AST sous forme de fonctions ou méthode tagless-final.


Maintenant que nous avons vu comment comment plonger un langage en OCaml selon la méthode initiale, nous traiterons une autre méthode — objet principal du présent journal — qui permet de résoudre un problème inaccessible à l'autre manière de procéder : étendre les langages en réutilisant le code déja écrit.

Langage additif et multiplicatif sur les entiers.


Le principe de base de cette méthode consistera à utiliser le système de modules du langage OCaml (en), de s'inspirer des GADT pour garantir le bon typage du langage et de la fonction fold pour déterminer le contenu des modules. Dans un premier temps, un module peut être vu comme un espace de noms contenant des déclarations de type et de valeurs. Chaque module dispose d'une signature, qui est l'équivalent des types pour les modules, et celle-ci exprime la nature des éléments déclarés par le module.


module type SIG = sig
  type t
  val i : t
end

module M : SIG = struct
  type t = int
  let i = 1
end

M.i;;
- : M.t = <abstr>


Dans cet exemple, un module satisfaisant la signature SIG doit déclarer un type t et une valeur i de type t. Ce que fait le module M définit ensuite, et l'on accède aux valeurs définies dans le module par la notation pointée. Ici c'est de peu d'utilité car comme extérieurement on ne sait rien sur le type t de la valeur, on ne peut pas en faire usage. Pour pouvoir l'utiliser, il faudrait que le module M définisse également des fonctions pour opérer sur les valeurs du type qu'il déclare. C'est ce que nous allons faire avec notre langage sur les entiers.


module type SYM_INT = sig
  type 'a repr
  type 'a obs

  val lit : int -> int repr
  val add : int repr -> int repr -> int repr
  val mul : int repr -> int repr -> int repr

  val observe : 'a repr -> 'a obs
end


Cette signature définit deux types et quelques fonctions pour opérer dessus. On peut remarquer que le type des fonctions lit, add et mul mime, sous forme curryfiée, les constructeurs du GADT de la première partie. Le type paramétré 'a repr servira à la représentation interne des termes de notre langage, tandis que le type 'a obs permettra de les présenter extérieurement. Le nom de la signature est préfixée par SYM ce qui renvoie au néologisme symantique forgée par les inventeurs de la méthode afin de signifier que celle-ci exprime à la fois la syntaxe et la sémantique du langage.


Pour implémenter un module satisfaisant cette signature, il suffit alors de réutiliser ce que nous avons fait précédemment pour diversifier les interprétations avec la fonction fold. Chaque module correspondra à une interprétation différente de notre langage, et le code des fonctions sera celui des arguments passés à la fonction fold. On se retrouve alors avec ces deux interprétations possibles :


module EvalInt = struct
  type 'a repr = 'a
  type 'a obs = 'a

  let lit n = n
  let add = ( + )
  let mul = ( * )

  let observe x = x
end

module ShowInt = struct
  type 'a repr = int -> string
  type 'a obs = string

  (* quelques fonctions utilitaires *)
  let paren s = "(" ^ s ^ ")"
  let cparen b = if b then paren else (fun s -> s)
  let show_int = string_of_int
  let show_add s s' = s ^ " + " ^ s'
  let show_mul s s' = s ^ " * " ^ s'

  let lit n = fun p -> show_int n
  let add x y = fun p ->
    cparen (p > 3) @@ show_add (x 3) (y 4)
  let mul x y = fun p ->
    cparen (p > 4) @@ show_mul (x 4) (y 5)

  let observe x = x 0
end


Afin de les tester, définissons un module d'exemples :


module ExInt (I : SYM_INT) = struct
  open I
  let observe = observe
  let ( + ) = add
  let ( * ) = mul
  let ex1 = lit 1 + lit 2
  let ex2 = (lit 1 + lit 2) * (lit 3 + lit 4)
  let ex3 = lit 1 + (lit 2 + lit 3 * lit 4)
end


Ce module, contrairement aux précédents, est paramétré par un autre module d'interprétation de notre langage : c'est ce que l'on appelle un foncteur. La directive open permet d'ouvrir l'espace de nom du module I pour ne pas avoir besoin d'utiliser la notation pointée. Pour l'usage, cela se passe comme suit :


(* on applique notre foncteur sur nos deux modules *)
module ExEval = ExInt(EvalInt)
module ExShow = ExInt(ShowInt)

(* on teste le bon fonctionnement de nos interprétations *)
ExEval.(observe ex2);;
- : int = 21

ExShow.(observe ex2);;
- : string = "(1 + 2) * (3 + 4)"

ExShow.(observe ex3);;
- : string = "1 + (2 + 3 * 4)"

Enrichissons le langage avec des booléens et conditions.


Comme dans la première partie nous allons maintenant tenter d'étendre notre langage, et c'est là que nous verrons la supériorité de cette méthode sur l'autre : on pourra réutiliser le code déjà écrit. On commence par définir la signature définissant la symantique du langage étendu.


module type SYM_INT_COND = sig
  include SYM_INT

  val bool_ : bool -> bool repr
  val if_ : bool repr -> (unit -> 'x) -> (unit -> 'x) -> ('a repr as 'x)
  val eq : int repr -> int repr -> bool repr
  val lt : int repr -> int repr -> bool repr
end


Afin d'étendre le langage, il suffit d'utiliser la directive include puis de rajouter ensuite les nouvelles constructions du langage. On en a profité ici pour ajouter des fonctions de comparaisons sur les entiers. Pour les nouvelles interprétations, on procède de même :


module EvalIntCond = struct
  include EvalInt

  let bool_ b = b
  let if_ b th el = if b then th () else el ()
  let eq = ( = )
  let lt = ( < )
end

module ShowIntCond = struct
  include ShowInt

  let show_bool = string_of_bool
  let show_if bs ts es = "if " ^ bs ^ " then " ^ ts ^ " else " ^ es
  let show_eq s s' = s ^ " = " ^ s'
  let show_lt s s' = s ^ " < "  ^ s'

  let bool_ b = fun p -> show_bool b
  let if_ b th el = fun p ->
    cparen (p > 0) @@ show_if (b 0) (th () 0) (el () 0)
  let eq x y = fun p ->
    cparen (p > 0) @@ show_eq (x 1) (y 1)
  let lt x y = fun p ->
    cparen (p > 0) @@ show_lt (x 1) (y 1)
end


On peut de même étendre notre module d'exemples pour tester ces nouvelles interprétations :


module ExIntCond (I : SYM_INT_COND) = struct
  include ExInt(I)
  open I
  let ( = ) = eq
  let ( < ) = lt

  let true_ = bool_ true
  let false_ = bool_ false

  let ex4 = if_ true_ (fun () -> ex3) (fun () -> ex2)
  let ex5 = if_ (lit 1 < lit 2) (fun () -> ex3) (fun () -> ex4)
  let ex6 = if_ (ex5 = ex3) (fun () -> true_) (fun () -> false_)
end


Quelques tests :


module ExEval = ExIntCond(EvalIntCond)
module ExShow = ExIntCond(ShowIntCond)

ExEval.(observe ex4);;
- : int = 15

ExEval.(observe ex5);;
- : int = 15

ExEval.(observe ex6);;
- : bool = true

ExShow.(observe ex6);;
- : string = 
"if (if 1 < 2 then 1 + (2 + 3 * 4)
    else if true then 1 + (2 + 3 * 4)
    else (1 + 2) * (3 + 4)) = 1 + (2 + 3 * 4)
then true else false"

ExShow.(observe ex4);;
- : string = "if true then 1 + (2 + 3 * 4) else (1 + 2) * (3 + 4)"

Fiat lux : Que la lumière soit, et la lumière fut !


Jusqu'ici nous avons représenté les termes de nos langages objets directement dans le langage hôte OCaml de deux manières distinctes :



	arbre syntaxique comme structure de données ;

	arbre syntaxique sous forme de fonctions.


Nous allons montrer qu'au fond la première technique n'est qu'un cas particulier de la seconde, et illustrer comment l'approche tagless-final généralise la première. Ainsi, la première se subsumant sous la seconde, dans la suite du journal nous n'utiliserons plus que cette dernière.


Reprenons le code écrit jusqu'ici pour la méthode avec structures de données et fold pour le langage additif et multiplicatif, puis encapsulons le dans un module.


module Ast = struct
  (* on y met les définitions du type, du fold
   * et des différentes intéprétations *)
end


À partir de là, rien de plus simple que de définir des modules de signature SYM_INT qui s'intègrent parfaitement dans l'approche tagless-final.


On commence par un module dont la représentation interne est justement notre structure de données :


module AstBase = struct
  type _ repr = Ast.expr
  let lit n = Ast.Lit n
  let add x y = Ast.Add (x,y)
  let mul x y = Ast.Mul (x,y)
end


puis on l'étend vers différents modules de signature SYM_INT qui observeront ou interpréteront nos arbres différemment.


module AstEval = struct
  include AstBase 
  type _ obs = 'a
  let observe = Ast.eval
end

module AstShow = struct
  include AstBase
  type _ obs = string
  let observe = Ast.show
end


On peut maintenant les utiliser sans problème avec notre module d'exemples :


module ExEval = ExInt(AstEval)
module ExShow = ExInt(AstShow)

ExEval.(observe ex3);;
- : int = 15

ExShow.(observe ex3);;
- : string = "1 + (2 + 3 * 4)"

Optimisons un peu nos interprètes !


Dans cette section nous présenterons un cadre général, et ses principes, pour effectuer des optimisations sur nos langages. Nous verrons aussi comment tirer profit de la structure modulaire du code pour réaliser cette tâche.

Suppression des opérandes nuls.


La première optimisation que l'on va voir est celle qui consiste à supprimer tous les litéraux nuls, en appliquant les règles : x + 0 = 0 + x = 0 et x * 0 = 0 * x = 0.


Pour ce faire, on partira d'un interprète I quelconque puis on représentera chaque terme à l'aide d'un GADT qui exprimera notre connaissance sur sa nullité. Enfin, pour l'observer, on se contentera de renvoyer l'observation par I.


module RemoveZero (I : SYM_INT) = struct
  type 'a repr =
    | Unk : 'a I.repr -> 'a repr
    | Zero : int repr
  type 'a obs = 'a I.obs

  let lit n = if n = 0 then Zero else Unk (I.lit n)
  let add x y = match x,y with
    | Zero, _ -> y
    | _, Zero -> x
    | Unk x, Unk y -> Unk (I.add x y)
  let mul x y = match x,y with
    | Zero, _
    | _, Zero -> Zero
    |Unk x, Unk y -> Unk (I.mul x y)

  let observe : type a. a repr -> a obs = function
    | Unk x -> I.observe x
    | Zero -> I.observe (I.lit 0)
end


On écrit un module de test pour vérifier que tout se passe bien :


module Test (I : SYM_INT) = struct
  open I
  let observe = observe
  let (+) = add
  let ( * ) = mul

  let e1 = lit 1 + lit 0 + lit 2
  let e2 = e1 + lit 0 * lit 3
end

(* sans optimisation *)
let module M = Test(ShowInt) in
M.(observe e2);;
- : string = "1 + 0 + 2 + 0 * 3"

(* l'optimiseur en pratique *)
let module M = Test(RemoveZero(ShowInt)) in
M.(observe e2);;
- : string = "1 + 2"

(* sur un interprète issu de la méthode avec variant *)
let module M = Test(RemoveZero(AstShow)) in
M.(observe e2);;
- : string = "1 + 2"

Principe général des transformations.


Lorsque l'on a supprimé les littéraux nuls, on a au fond transformer l'arbre syntaxique (comme le montre leur représentation sous forme de chaîne de caractères) mais sans modifier son interprétation naturelle en tant qu'entier.


À cet effet, on est parti d'une représentation donnée par un interprète I, puis l'on a envoyé les représentations dans un autre domaine Unk : 'a I.repr -> 'a repr | Zero : int repr afin d'opérer dessus, pour enfin revenir dans le domaine de l'interpréteur I et observer la transformation. Le principe de l'aller-retour entre deux domaines est capturé par la signature de module suivante :


module type TRANS = sig
  type 'a from
  type 'a term
  val fwd : 'a from -> 'a term
  val bwd : 'a term -> 'a from
end


L'optimiseur travaille sur des 'a term qui contiennent les informations nécessaires à sa tâche. Ils sont obtenus à partir d'une autre représentation intermédiaire 'a from via la fonction fwd puis, une fois l'optimisation terminée, on revient dans la représentation de départ à l'aide de la fonction bwd.


On peut coder à partir d'une transformation X : TRANS un « optimiseur » trivial qui… ne change rien :-P


module SymIntT (X : TRANS) (I : SYM_INT with type 'a repr = 'a X.from) = struct
  open X
  type 'a repr = 'a term
  type 'a obs = 'a I.obs

  let lit n = fwd @@ I.lit n
  let add x y = fwd @@ I.add (bwd x) (bwd y)
  let mul x y = fwd @@ I.mul (bwd x) (bwd y)
  let observe x = I.observe (bwd x)
end


Pour avoir un optimiseur plus utile que celui-ci, il faut fournir deux choses :



	une transformation X : TRANS pour obtenir des termes sur lesquels opérer ;

	un interprète partiel qui opère sur la partie du langage qu'il optimise.


Le plus simple est de mettre le tout dans un module, ce qui dans le cas de la suppression des zéros donne :


module RemoveZeroPass (I : SYM_INT) = struct
  (* la transformation pour faire l'aller-retour entre les deux domaines *)
  module X = struct
    type 'a from = 'a I.repr
    type 'a term =
      | Unk : 'a I.repr -> 'a term
      | Zero : int term

    let fwd x = Unk x
    let bwd : type a. a term -> a from = function
      | Unk x -> x
      | Zero -> I.lit 0
  end
  open X 
  (* la passe d'optimisation proprement dite *)
  module IDelta = struct
    let lit n = if n = 0 then Zero else fwd @@ I.lit n
    let add x y = match x,y with
      | Zero, _ -> y
      | _, Zero -> x
      | _ -> fwd @@ I.add (bwd x) (bwd y)
    let mul x y = match x,y with
      | Zero, _
      | _, Zero -> Zero
      | _ -> fwd @@ I.mul (bwd x) (bwd y)
  end
end


Afin de préserver l'interprétation des termes qui ne sont pas concernés par la passe d'optimisation, il faut s'assurer que l'on a toujours bwd (fwd x) = x, ce qui est évident dans notre cas. Le module IDelta porte ce nom pour signifier qu'il contient la différence (le delta) d'interprétation par rapport à l'interprète initial I.


À partir de là, on obtient l'optimiseur de la manière suivante :


module RemoveZeroInt (I : SYM_INT) = struct
  module OptM = RemoveZeroPass(I)

  (* on inclut l'optimiseur trivial qui ne fait rien *)
  include SymIntT(OptM.X)(I)

  (* on surcharge les fonctions lit, add et mul *)
  include OptM.IDelta
end

let module M = Test(RemoveZeroInt(ShowInt)) in
M.(observe e2);;
- : string = "1 + 2"


Pour le lecteur qui aime bien les graphiques, on pourrait représenter la passe d'optimisation ainsi :


Domaine 'a from : terme t         t optimisé ------> t observé : 'a obs
                      |                 ^   I.observe
                      |                 |
                  fwd |                 | bwd
                      |                 |
                      V    passe        |
Domaine 'a term : terme t' -----> t' optimisé



Cette façon de procéder peut sembler étrange au premier abord : pourquoi inclure un optimiseur qui ne fait rien, pour ensuite remplacer immédiatement les fonctions qu'il importe ? Cela n'a effectivement aucun intérêt dans ce cas précis, mais elle illustre un principe général qui sera fort apprécié lorsque la passe ne modifiera pas toutes les constructions du langage : seules les fonctions de l'optimiseur trivial réellement utilisées par le notre seront remplacées et les autres resteront inchangées.

Simplification de la multiplication par un.


Dans la même lignée que l'optimisation précédente, on peut simplifier nos termes du langage en faisant usage de la règle : x * 1 = 1 * x = x.


module RemoveOnePass (I : SYM_INT) = struct
  module X = struct
    type 'a from = 'a I.repr
    type 'a term =
      | Unk : 'a from -> 'a term
      | One : int term
    let fwd x = Unk x
    let bwd : type a. a term -> a from = function
      | Unk x -> x
      | One -> I.lit 1
  end
  open X
  module IDelta = struct
    let lit n = if n = 1 then One else fwd @@ I.lit n
    let mul x y = match x,y with
      | One, _ -> y
      | _, One -> x
      | _ -> fwd @@ I.mul (bwd x) (bwd y)
  end
end

(* on utilise la passe comme au-dessus *)
module RemoveOneInt (I : SYM_INT) = struct
  module OptM = RemoveOnePass(I)
  include SymIntT(OptM.X)(I)
  (* on surcharge lit et mul *)
  include OptM.IDelta
end


Ici la transformation triviale est utile car la passe ne touche pas à l'addition. Un petit test qui combine les deux passes :


module Test (I : SYM_INT) = struct
  open I
  let observe = observe
  let ( +) = add
  let ( * ) = mul

  let e = lit 1 * lit 2 + lit 0
end

let module M = Test(RemoveZeroInt(ShowInt)) in
M.(observe e);;
- : string = "1 * 2"

let module M = Test(RemoveOneInt(ShowInt)) in
M.(observe e);;
- : string = "2 + 0"

(* nos deux passes commutent *)
let module M = Test(RemoveOneInt(RemoveZeroInt(ShowInt))) in
M.(observe e);;
- : string = "2"

let module M = Test(RemoveZeroInt(RemoveOneInt(ShowInt))) in
M.(observe e);;
- : string = "2"

Réutilisation de nos passes sur le langage étendu.


Un des grands avantages de la méthode tagless-final est de pouvoir réutiliser du code déjà écrit en structurant le programme de façon modulaire. Les optimisations n'échappent pas à la règle. Pour récupérer nos précédentes passes, il faut d'abord définir notre transformation triviale pour le langage étendu.


module SymIntCondT (X : TRANS)
(I : SYM_INT_COND with type 'a repr = 'a X.from) = struct
  (* on commence par récupérer la précédente *)
  include SymIntT (X)(I)
  open X  
  (* on ajoute les nouvelles constructions du langage *)
  let bool_ b = fwd @@ I.bool_ b
  let if_ b th el =
    fwd @@ I.if_ (bwd b) (fun () -> bwd @@ th ()) (fun () -> bwd @@ el ())
  let eq x y = fwd @@ I.eq (bwd x) (bwd y)
  let lt x y = fwd @@ I.lt (bwd x) (bwd y)
end


Et l'on peut alors réutiliser nos passes directement :


module RemoveZeroIntCond (I : SYM_INT_COND) = struct
  module OptM = RemoveZeroPass(I)
  include SymIntCondT(OptM.X)(I)
  (* on ne surcharge que les constructeurs optimisés *)
  include OptM.IDelta
end

module RemoveOneIntCond (I : SYM_INT_COND) = struct
  module OptM = RemoveOnePass(I)
  include SymIntCondT(OptM.X)(I)
  (* on ne surcharge que les constructeurs optimisés *)
  include OptM.IDelta
end


Un petit test pour la route :


module Test (I : SYM_INT_COND) = struct
  open I
  let observe = observe
  let ( + ) = add
  let ( * ) = mul
  let ( = ) = eq

  let e =
    if_ (lit 1 * lit 2 = lit 2 + lit 0)
    (fun () -> lit 3 * lit 1)
    (fun () -> lit 4 + lit 0)
end

let module M = Test(RemoveZeroIntCond(ShowIntCond)) in
M.(observe e);;
- : string = "if 1 * 2 = 2 then 3 * 1 else 4"

let module M = Test(RemoveOneIntCond(ShowIntCond)) in
M.(observe e);;
- : string = "if 2 = 2 + 0 then 3 else 4 + 0"

let module M = Test(RemoveOneIntCond(RemoveZeroIntCond(ShowIntCond))) in
M.(observe e);;
- : string = "if 2 = 2 then 3 else 4"

Étendre le langage avec des fonctions.


Notre langage commence à devenir sympa, mais ce serait mieux si on lui rajoutait la possibilité de définir des fonctions. C'est ce que nous ferons dans cette dernière partie, et nous rajouterons une passe de calcul des valeurs statiquement connues.

La symantique étendue et ses interprètes.


Pour étendre la signature du langage et rajouter des fonctions, rien de plus simple :


module type SYM_INT_COND_HO = sig
  include SYM_INT_COND
  val lam : ('a repr -> 'b repr) -> ('a -> 'b) repr
  val app : ('a -> 'b) repr -> 'a repr -> 'b repr
end


On ajoute un opérateur d'abstraction lam pour créer des fonctions, et un opérateur d'application app pour les utiliser. On étend ensuite nos interprètes en commençant par l'évaluateur.


module EvalIntCondHO = struct
  include EvalIntCond
  let lam f = f
  let app f x = f x
end


Pour l'interprète de pretty printing, il va falloir adapter légèrement la représentation interne pour prendre en compte le nombre de paramètres des fonctions.


module ShowIntCondHO = struct
  (*
    on ajoute un paramètre entier à nos représentations
    pour prendre en compte le nombre de variables de nos
    fonctions et éviter les colisions de noms.
  *)
  type 'a repr = int -> int -> string
  type 'a obs = string

  (*
    on surcharge nos fonctions de l'interprète du sous-langage
    avec entiers et booléens que l'on enveloppe pour prendre
    en compte le nouveau paramètre.
  *)
  open ShowIntCond

  let lit n = fun c -> lit n
  let add x y = fun c -> add (x c) (y c)
  let mul x y = fun c -> mul (x c) (y c)
  let bool_ b = fun c -> bool_ b
  let eq e e' = fun c -> eq (e c) (e' c)
  let lt i j = fun c -> lt (i c) (j c)
  let if_ b then_e else_e = fun c ->
    if_ (b c) (fun () -> then_e () c) (fun () -> else_e () c)

  (* quelques fonctions utilitaires *)
  let show_lam v body = "fun " ^ v ^ " -> " ^ body
  let varnames = "xyztuvw"
  let varname = function
    | i when i < String.length varnames ->
        String.make 1 varnames.[i]
    | i -> "x" ^ string_of_int i

  let lam f = fun c p ->
    let v = varname c in
    let body = f (fun _ _-> v) (c + 1) 0 in
    cparen (p > 0) @@ show_lam v body

  let app f x = fun c p -> 
    cparen (p > 10) (f c 10 ^ " " ^ x c 11)

  let observe x = x 0 0
end


On peut toujours interpréter les précédents exemples avec ces nouveaux modules :


let module M = ExIntCond(EvalIntCondHO) in
M.(observe ex3);;
- : int = 15

let module M = ExIntCond(ShowIntCondHO) in
M.(observe ex3);;
- : string = "1 + (2 + 3 * 4)"


Mais également des termes du nouveau langage, comme il se doit :


module ExHO (I : SYM_INT_COND_HO) = struct
  include ExIntCond(I)
  open I
  let app = app
  let lit = lit
  let ex7 = lam @@ fun x ->
    (x + lit 2 + lit 3 * x) * (lit 2 + x)
  let ex8 = app ex7 (lit 3)
  let ex9 x = lam @@ fun y ->
    if_ (y < lit x) (fun () -> lit 2 * y) (fun () -> lit 3 * y)
end

let module M = ExHO(EvalIntCondHO) in
M.(observe ex7);;
- : int -> int = <fun>

let module M = ExHO(EvalIntCondHO) in
M.(observe ex7) 3;;
- : int = 70

let module M = ExHO(EvalIntCondHO) in
M.(observe ex8);;
- : int = 70

let module M = ExHO(ShowIntCondHO) in
M.(observe ex7);;
- : string = "fun x -> (x + 2 + 3 * x) * (2 + x)"

let module M = ExHO(ShowIntCondHO) in
M.(observe ex8);;
- : string = "(fun x -> (x + 2 + 3 * x) * (2 + x)) 3"

let module M = ExHO(StaticHO(ShowIntCondHO)) in
M.(observe (ex9 2));;
- : string = "fun x -> if x < 2 then 2 * x else 3 * x"

Évaluation partielle et statique.


Il ne nous reste plus qu'à réaliser une optimisation qui réduit ce qui est statiquement connu. On commencera par définir notre transformation triviale puis nous passerons à la passe proprement dite.


module SymIntCondHOT (X : TRANS)
(I : SYM_INT_COND_HO with type 'a repr = 'a X.from) = struct
  include SymIntCondT(X)(I)
  open X
  let lam f = fwd @@ I.lam @@ fun x -> bwd (f (fwd x))
  let app f x = fwd @@ I.app (bwd f) (bwd x)
end


Pour la passe statique le code est assez simple : on commence par étiquetés nos termes pour savoir s'il sont statiquement connus (Sta), si c'est une fonction (Fun) ou s'ils ne seront connus que dynamiquement (Dyn). Dans le cas des fonctions, on les applique statiquement, si l'on opère avec des valeurs statiques alors on effectue le calcul statiquement, sinon on se rabat sur notre interprète I.


module StaticPass (I : SYM_INT_COND_HO) = struct
  module X = struct
    type 'a from = 'a I.repr    
    type 'a term =
      | Dyn : 'a from -> 'a term
      | Sta : ('a * 'a from) -> 'a term
      | Fun : ('a term -> 'b term) -> ('a -> 'b) term

    let fwd x = Dyn x
    let rec bwd : type a. a term -> a from = function
      | Dyn x -> x
      | Sta (_,x) -> x
      | Fun f -> I.lam @@ fun x -> bwd (f (fwd x))
  end
  open X
  module IDelta = struct
    let lit n = Sta (n, I.lit n)
    let add x y = match x,y with
      | Sta (x,_) , Sta (y,_) -> lit (x + y)
      | _ -> fwd @@ I.add (bwd x) (bwd y)
    let mul x y = match x,y with
      | Sta (x,_) , Sta (y,_) -> lit (x * y)
      | _ -> fwd @@ I.mul (bwd x) (bwd y)

    let bool_ b = Sta (b, I.bool_ b)
    let if_ b th el = match b with
      | Sta (b,_) -> if b then th () else el ()
      | _ -> fwd @@ I.if_ (bwd b) (fun () -> bwd @@ th()) (fun () -> bwd @@ el())

    let eq x y = match x,y with
      | Sta (x,_) , Sta(y,_) -> bool_ (x = y)
      | _ -> fwd @@ I.eq (bwd x) (bwd y)

    let lt x y = match x,y with
      | Sta (x,_) , Sta (y,_) -> bool_ ( x < y )
      | _ -> fwd @@ I.lt (bwd x) (bwd y)

    let lam f = Fun f

    let app f x = match f with
      | Fun f -> f x
      | _ -> fwd @@ I.app (bwd f) (bwd x)
  end
end


On peut alors définir notre optimiseur et récupérer les précédents.


module RemoveZeroHO (I : SYM_INT_COND_HO) = struct
  module OptM = RemoveZeroPass(I)
  include SymIntCondHOT(OptM.X)(I)
  include OptM.IDelta
end

module RemoveOneHO (I : SYM_INT_COND_HO) = struct
  module OptM = RemoveOnePass(I)
  include SymIntCondHOT(OptM.X)(I)
  include OptM.IDelta
end

module StaticHO (I : SYM_INT_COND_HO) = struct
  module OptM = StaticPass(I)
  include SymIntCondHOT(OptM.X)(I)
  include OptM.IDelta
end


La passe statique en action :


let module M = ExHO(ShowIntCondHO) in
M.(observe ex8);;
- : string = "(fun x -> (x + 2 + 3 * x) * (2 + x)) 3"

let module M = ExHO(StaticHO(ShowIntCondHO)) in
M.(observe ex8);;
- : string = "70"

let module M = ExHO(ShowIntCondHO) in
M.(observe (ex9 4));;
- : string = "fun x -> if x < 4 then 2 * x else 3 * x"

let module M = ExHO(ShowIntCondHO) in
M.(observe @@ app (ex9 4) (lit 2));;
- : string = "(fun x -> if x < 4 then 2 * x else 3 * x) 2"

let module M = ExHO(StaticHO(ShowIntCondHO)) in
M.(observe @@ app (ex9 4) (lit 2));;
- : string = "4"

let module M = ExHO(StaticHO(ShowIntCondHO)) in
M.(observe @@ app (ex9 1) (lit 2));;
- : string = "6"

Conclusion.


Le journal n'a présenté que dans les grands traits la méthode dite tagless-final pour plonger un langage dans un langage fonctionnel hôte. Comparée à la méthode plus classique consistant à représenter l'arbre de syntaxe abstrait du langage par un type de données algébriques (généralisé ou non), celle-ci permet d'étendre facilement les langages sans avoir à réécrire de code. De plus, elle garantie la bonne formation (au niveau du typage) des termes de nos langages en réutilisant le système de types du langage hôte : pas besoin d'écrire un type checker. De par sa structure modulaire, le code permet d'écrire des passes d'optimisations qui se concentrent uniquement sur le sous-ensemble du langage à optimiser. Par construction, ces dernières sont assurées de préserver le typage des termes et en « collant » à la sémantique dénotanionnelle du langage il est plus aisé de s'assurer de leur bon fonctionnement.


Les exemples du journal sont surtout des exemples jouets, ils servent essentiellement à illustrer les principes généraux de la méthode tagless-final. Pour un exemple d'application, sans doute plus utile dans la vie courante d'un développeur, on pourra regarder du côté de quel (sous licence MIT). Il permet de générer des requêtes SQL optimisées avec la garantie du typage statique. Son fonctionnement est présenté dans l'article Finally, Safely-Extensible and Efficient Language-Integrated Query.


On pourra également consulter l'exemple du langage des circuits logiques. Il se trouve sur la page du tutoriel sur l'optimisation via la méthode tagless-final de Oleg Kiselyov, tutoriel qui m'a servi de base pour ce journal. On y trouve des exemples de codes en Haskell et OCaml.


Comme référence, il y a enfin l'article originel présentant la méthode : Finally Tagless, Partially Evaluated. Celle-ci y est combinée à MetaOCaml (présenté récemment dans une dépêche)  afin d'obtenir un interprète, un compilateur, un évaluateur partiel ou encore différentes transformations ou méthodes d'évaluation (par nom, par valeur…) pour le langage objet.

